

NAG Fortran Library Routine Document

F07CUF (ZGTCON)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07CUF (ZGTCON) estimates the reciprocal condition number of a complex n by n tridiagonal matrix A , using the LU factorization returned by F07CRF (ZGTTRF).

2 Specification

```

SUBROUTINE F07CUF (NORM, N, DL, D, DU, DU2, IPIV, ANORM, RCOND, WORK,
1                INFO)
    INTEGER          N, IPIV(*), INFO
    double precision ANORM, RCOND
    complex*16      DL(*), D(*), DU(*), DU2(*), WORK(*)
    CHARACTER*1     NORM

```

The routine may be called by its LAPACK name *zgtcon*.

3 Description

F07CUF (ZGTCON) should be preceded by a call to F07CRF (ZGTTRF), which uses Gaussian elimination with partial pivoting and row interchanges to factorize the matrix A as

$$A = PLU,$$

where P is a permutation matrix, L is unit lower triangular with at most one non-zero subdiagonal element in each column, and U is an upper triangular band matrix, with two superdiagonals. F07CUF (ZGTCON) then utilizes the factorization to estimate either $\|A^{-1}\|_1$ or $\|A^{-1}\|_\infty$, from which the estimate of the reciprocal of the condition number of A , $1/\kappa(A)$ is computed as either

$$1/\kappa_1(A) = 1/(\|A\|_1\|A^{-1}\|_1)$$

or

$$1/\kappa_\infty(A) = 1/(\|A\|_\infty\|A^{-1}\|_\infty).$$

$1/\kappa(A)$ is returned, rather than $\kappa(A)$, since when A is singular $\kappa(A)$ is infinite.

Note that $\kappa_\infty(A) = \kappa_1(A^T)$.

4 References

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

5 Parameters

1: NORM – CHARACTER*1 *Input*
On entry: specifies the norm to be used to estimate $\kappa(A)$.
 NORM = '1' or 'O'
 Estimate $\kappa_1(A)$.

- NORM = 'I'
 Estimate $\kappa_{\infty}(A)$.
 Constraint: NORM = '1', 'O' or 'I'.
- 2: N – INTEGER *Input*
 On entry: n , the order of the matrix A .
 Constraint: $N \geq 0$.
- 3: DL(*) – **complex*16** array *Input*
Note: the dimension of the array DL must be at least $\max(1, N - 1)$.
 On entry: must contain the $(n - 1)$ multipliers that define the matrix L of the LU factorization of A .
- 4: D(*) – **complex*16** array *Input*
Note: the dimension of the array D must be at least $\max(1, N)$.
 On entry: must contain the n diagonal elements of the upper triangular matrix U from the LU factorization of A .
- 5: DU(*) – **complex*16** array *Input*
Note: the dimension of the array DU must be at least $\max(1, N - 1)$.
 On entry: must contain the $(n - 1)$ elements of the first superdiagonal of U .
- 6: DU2(*) – **complex*16** array *Input*
Note: the dimension of the array DU2 must be at least $\max(1, N - 2)$.
 On entry: must contain the $(n - 2)$ elements of the second superdiagonal of U .
- 7: IPIV(*) – INTEGER array *Input*
Note: the dimension of the array IPIV must be at least $\max(1, N)$.
 On entry: must contain the n pivot indices that define the permutation matrix P . At the i th step, row i of the matrix was interchanged with row $IPIV(i)$, and $IPIV(i)$ must always be either i or $(i + 1)$, $IPIV(i) = i$ indicating that a row interchange was not performed.
- 8: ANORM – **double precision** *Input*
 On entry: if NORM = '1' or 'O', ANORM must contain $\|A\|_1$.
 If NORM = 'I', ANORM must contain $\|A\|_{\infty}$.
 ANORM may be computed by calling F06UNF with the same value for NORM.
 ANORM must be computed either before calling F07CRF (ZGTTRF), or else from a copy of the original matrix A .
- 9: RCOND – **double precision** *Output*
 On exit: contains an estimate of the reciprocal condition number.
- 10: WORK(*) – **complex*16** array *Workspace*
Note: the dimension of the array WORK must be at least $\max(1, 2 \times N)$.
- 11: INFO – INTEGER *Output*
 On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = $-i$, the i th argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

In practice the condition number estimator is very reliable, but it can underestimate the true condition number; see Section 15.3 of Higham (2002) for further details.

8 Further Comments

The condition number estimation typically requires between four and five solves and never more than eleven solves, following the factorization. The total number of floating-point operations required to perform a solve is proportional to n .

The real analogue of this routine is F07CGF (DGTCON).

9 Example

To estimate the condition number in the 1-norm of the tridiagonal matrix A given by

$$A = \begin{pmatrix} -1.3 + 1.3i & 2.0 - 1.0i & 0 & 0 & 0 \\ 1.0 - 2.0i & -1.3 + 1.3i & 2.0 + 1.0i & 0 & 0 \\ 0 & 1.0 + 1.0i & -1.3 + 3.3i & -1.0 + 1.0i & 0 \\ 0 & 0 & 2.0 - 3.0i & -0.3 + 4.3i & 1.0 - 1.0i \\ 0 & 0 & 0 & 1.0 + 1.0i & -3.3 + 1.3i \end{pmatrix}.$$

9.1 Program Text

```
*      F07CUF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=50)
*      .. Local Scalars ..
      DOUBLE PRECISION ANORM, RCOND
      INTEGER          I, INFO, N
*      .. Local Arrays ..
      COMPLEX *16      D(NMAX), DL(NMAX-1), DU(NMAX-1), DU2(NMAX-2),
+                     WORK(2*NMAX)
      INTEGER          IPIV(NMAX)
*      .. External Functions ..
      DOUBLE PRECISION F06UNF, X02AJF
      EXTERNAL         F06UNF, X02AJF
*      .. External Subroutines ..
      EXTERNAL         ZGTCON, ZGTTRF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F07CUF Example Program Results'
      WRITE (NOUT,*)
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*          Read the tridiagonal matrix A from data file
*
*
      READ (NIN,*) (DU(I),I=1,N-1)
      READ (NIN,*) (D(I),I=1,N)
```

```

      READ (NIN,*) (DL(I),I=1,N-1)
*
*   Compute the 1-norm of A
*
      ANORM = F06UNF('1-norm',N,DL,D,DU)
*
*   Factorize the tridiagonal matrix A
*
      CALL ZGTTRF(N,DL,D,DU,DU2,IPIV,INFO)
*
      IF (INFO.EQ.0) THEN
*
*       Estimate the condition number of A
*
*       CALL ZGTCON('1-norm',N,DL,D,DU,DU2,IPIV,ANORM,RCOND,WORK,
+           INFO)
*
*       Print the estimated condition number
*
      IF (RCOND.GE.X02AJF()) THEN
+         WRITE (NOUT,99999) 'Estimate of condition number = ',
+           1.0D0/RCOND
      ELSE
+         WRITE (NOUT,99999)
+           'A is singular to working precision. RCOND = ', RCOND
      END IF
*
      ELSE
+         WRITE (NOUT,99998) 'The (', INFO, ',', INFO, ')',
+           ' element of the factor U is zero'
      END IF
      ELSE
      WRITE (NOUT,*) 'NMAX too small'
      END IF
      STOP
*
99999 FORMAT (1X,A,1P,E10.2)
99998 FORMAT (1X,A,I3,A,I3,A,A)
      END

```

9.2 Program Data

F07CUF Example Program Data

```

5                                     :Value of N
( 2.0,-1.0) ( 2.0, 1.0) (-1.0, 1.0) ( 1.0,-1.0) :End of DU
(-1.3, 1.3) (-1.3, 1.3) (-1.3, 3.3) (-0.3, 4.3)
(-3.3, 1.3)                                     :End of D
( 1.0,-2.0) ( 1.0, 1.0) ( 2.0,-3.0) ( 1.0, 1.0) :End of DL

```

9.3 Program Results

F07CUF Example Program Results

```

Estimate of condition number = 1.84E+02

```
